The Concentration Index

Concentration curves can be used to identify whether socioeconomic inequality in some health sector variable exists and whether it is more pronounced at one point in time than another or in one country than another. But a concentration curve does not give a measure of the magnitude of inequality that can be compared conveniently across many time periods, countries, regions, or whatever may be chosen for comparison. The concentration index (Kakwani 1977, 1980), which is directly related to the concentration curve, does quantify the degree of socioeconomic-related inequality in a health variable (Kakwani, Wagstaff, and van Doorslaer 1997; Wagstaff, van Doorslaer, and Paci 1989). It has been used, for example, to measure and to compare the degree of socioeconomic-related inequality in child mortality (Wagstaff 2000), child immunization (Gwatkin et al. 2003), child malnutrition (Wagstaff, van Doorslaer, and Watanabe 2003), adult health (van Doorslaer et al. 1997), health subsidies (O’Donnell et al. 2007), and health care utilization (van Doorslaer et al. 2006). Many other applications are possible.

In this chapter we define the concentration index, comment on its properties, and identify the required measurement properties of health sector variables to which it can be applied. We also describe how to compute the concentration index and how to obtain a standard error for it, both for grouped data and for microdata.

Definition and properties

Definition

The concentration index is defined with reference to the concentration curve, introduced in chapter 7. The concentration index is defined as twice the area between the concentration curve and the line of equality (the 45-degree line). So, in the case in which there is no socioeconomic-related inequality, the concentration index is zero. The convention is that the index takes a negative value when the curve lies above the line of equality, indicating disproportionate concentration of the health variable among the poor, and a positive value when it lies below the line of equality. If the health variable is a “bad” such as ill health, a negative value of the concentration index means ill health is higher among the poor.

Formally, the concentration index is defined as

\[
C = 1 - 2 \int_0^1 L(p)p dp .
\]
The index is bounded between –1 and 1. For a discrete living standards variable, it can be written as

\[
C = \frac{2}{N\mu} \sum_{i=1}^{N} h_i \eta_i - 1 - \frac{1}{N},
\]

where \( h_i \) is the health sector variable, \( \mu \) is its mean, and \( \eta_i = i/N \) is the fractional rank of individual \( i \) in the living standards distribution, with \( i = 1 \) for the poorest and \( i = N \) for the richest.\(^1\) For computation, a more convenient formula for the concentration index defines it in terms of the covariance between the health variable and the fractional rank in the living standards distribution (Jenkins 1988; Kakwani 1980; Lerman and Yitzhaki 1989),

\[
C = \frac{2}{\mu} \text{cov}(h, r).
\]

Note that the concentration index depends only on the relationship between the health variable and the rank of the living standards variable and not on the variation in the living standards variable itself. A change in the degree of income inequality need not affect the concentration index measure of income-related health inequality.

The concentration index summarizes information from the concentration curve and can do so only through the imposition of value judgments about the weight given to inequality at different points in the distribution. Alternative weighting schemes implying different judgments about attitudes to inequality are considered in the next chapter. Inevitably, the concentration index loses some of the information that is contained in the concentration curve. The index can be zero either because the concentration curve lies everywhere on top of the 45-degree line or because it crosses the line and the (weighted) areas above and below the line cancel out. It is obviously important to distinguish between such cases, and so the summary index should be examined in conjunction with the concentration curve.

The sign of the concentration index indicates the direction of any relationship between the health variable and position in the living standards distribution, and its magnitude reflects both the strength of the relationship and the degree of variability in the health variable. Although this is valuable information, one may also wish to place an intuitive interpretation on the value of the index. Koolman and van Doorslaer (2004) have shown that multiplying the value of the concentration index by 75 gives the percentage of the health variable that would need to be (linearly) redistributed from the richer half to the poorer half of the population (in the case that health inequality favors the rich) to arrive at a distribution with an index value of zero.

Properties

The properties of the concentration index depend on the measurement characteristics of the variable of interest. Strictly, the concentration index is an appropriate measure of socioeconomic-related health (care) inequality when health (care) is measured on a ratio scale with nonnegative values. The concentration index is

\(^1\)For large \( N \), the final term in equation 8.3 approaches zero and it is often omitted.
The Concentration Index

invariant to multiplication of the health sector variable of interest by any scalar (Kakwani 1980). So, for example, if we are measuring inequality in payments for health care, it does not matter whether payments are measured in local currency or in dollars; the concentration index will be the same. Similarly, it does not matter whether health care is analyzed in terms of utilization per month or if monthly data are multiplied by 12 to give yearly figures. However, the concentration index is not invariant to any linear transformation of the variable of interest. Adding a constant to the variable will change the value of the concentration index. In many applications this does not matter because there is no reason to make an additive transformation of the variable of interest. There is one important application in which this does represent a limitation, however. We are often interested in inequality in a health variable that is not measured on a ratio scale. A ratio scale has a true zero, allowing statements such as “A has twice as much X as B.” That makes sense for dollars or height. But many aspects of health cannot be measured in this way. Measurement of health inequality often relies on self-reported indicators of health, such as those considered in chapter 5. A concentration index cannot be computed directly from such categorical data. Although the ordinal data can be transformed into some cardinal measure and a concentration index computed for this (van Doorslaer and Jones 2003; Wagstaff and van Doorslaer 1994), the value of the index will depend on the transformation chosen (Erreygers 2005). In cross-country comparisons, even if all countries adopt the same transformation, their ranking by the concentration index could be sensitive to differences in the means of health that are used in the transformation.

A partial solution to this problem would be to dichotomize the categorical health measure. For example, one could examine how the proportion of individuals reporting poor health varies with living standards. Unfortunately, this introduces another problem. Wagstaff (2005) has demonstrated that the bounds of the concentration index for a dichotomous variable are not –1 and 1 but depend on the mean of the variable. For large samples, the lower bound is \(\mu - 1\) and the upper bound is \(1 - \mu\). So the feasible interval of the index shrinks as the mean rises. One should be cautious, therefore, in using the concentration index to compare inequality in, for example, child mortality and immunization rates across countries with substantial differences in the means of these variables. An obvious response is to normalize the concentration index by dividing through by 1 minus the mean (Wagstaff 2005).

If the health variable of interest takes negative as well as positive values, then its concentration index is not bounded within the range of (–1,1). In the extreme, if the mean of the variable were 0, the concentration index would not be defined.

Bleichrodt and van Doorslaer (2006) have derived the conditions that must hold for the concentration index (and related measures) to be a measure of socioeconomic-related health inequality consistent with a social welfare function. They argue that one condition—the principle of income-related health transfers—is rather restrictive. Erreygers (2006) has derived an alternative measure of socioeconomic-related health inequality that is consistent with this condition and three others argued to be desirable.

\(^2\)Erreygers (2005) suggests a couple of alternatives to the concentration index to deal with this problem.
Estimation and inference for grouped data

Point estimate of the concentration index

The concentration index for \( t=1,\ldots,T \) groups is easily computed in a spreadsheet program using the following formula (Fuller and Lury 1977):

\[
C = (p_1L_2 - p_2L_1) + (p_2L_3 - p_3L_2) + \ldots + (p_{T-1}L_T - p_TL_{T-1})
\]

where \( p_t \) is the cumulative percentage of the sample ranked by economic status in group \( t \), and \( L_t \) is the corresponding concentration curve ordinate. To illustrate, consider the distribution of under-five mortality by wealth quintiles in India, 1982–92. We drew the concentration curve for these data in chapter 7. Table 8.1 reproduces table 7.1 with the terms in brackets in the formula above added to the final column. The sum of these terms is –0.1694, which is the concentration index. The negative concentration index reflects the higher mortality rates among poorer children.

Standard error

A standard error of the estimator of \( C \) in the grouped data case can be computed using a formula given in Kakwani, Wagstaff, and van Doorslaer (1997). Let \( f_t \) be the proportion of the sample in the \( t \)th group, and define the fractional rank of group \( t \) by

\[
R_t = \sum_{k=1}^{t-1} f_k + \frac{1}{2} f_t,
\]

which is the cumulative proportion of the population up to the midpoint of each group interval. The variance of the estimator of \( C \) is given by

\[
\text{var}(\hat{C}) = \frac{1}{n} \left[ \sum_{t=1}^{T} f_t a_t^2 - (1 + C)^2 \right] + \frac{1}{n \mu^2} \sum_{t=1}^{T} f_t \sigma_t^2 (2R_t - 1 - C)^2,
\]

where \( n \) is the sample size, \( \sigma_t^2 \) is the variance of the health variable in the \( t \)th group, \( \mu \) is its mean,

\[
a_t = \mu / (2R_t - 1 - C) + 2 - q_{t-1} - q_t, \quad \text{and}
\]

\[
q_t = \frac{1}{\mu} \sum_{k=1}^{t-1} \mu_k f_k,
\]

which is the ordinate of \( L_h(p) \), \( q_0 = 0 \), and \( p_t = \sum_{k=1}^{t} f_k R_k \) (Kakwani, Wagstaff, and van Doorslaer 1997).

Table 8.1 Under-Five Deaths in India, 1982–92

<table>
<thead>
<tr>
<th>Wealth group</th>
<th>No. of births</th>
<th>Rel % births</th>
<th>Cumul % births</th>
<th>U5MR per 1,000</th>
<th>No. of deaths</th>
<th>Rel % deaths</th>
<th>Cumul % deaths</th>
<th>Conc. index</th>
</tr>
</thead>
<tbody>
<tr>
<td>Poorest</td>
<td>29,939</td>
<td>23</td>
<td>23</td>
<td>154.7</td>
<td>4,632</td>
<td>30</td>
<td>30</td>
<td>–0.0008</td>
</tr>
<tr>
<td>2nd</td>
<td>28,776</td>
<td>22</td>
<td>45</td>
<td>152.9</td>
<td>4,400</td>
<td>29</td>
<td>59</td>
<td>–0.0267</td>
</tr>
<tr>
<td>Middle</td>
<td>26,528</td>
<td>20</td>
<td>66</td>
<td>119.5</td>
<td>3,170</td>
<td>21</td>
<td>79</td>
<td>–0.0592</td>
</tr>
<tr>
<td>4th</td>
<td>24,689</td>
<td>19</td>
<td>85</td>
<td>86.9</td>
<td>2,145</td>
<td>14</td>
<td>93</td>
<td>–0.0827</td>
</tr>
<tr>
<td>Richest</td>
<td>19,739</td>
<td>15</td>
<td>100</td>
<td>54.3</td>
<td>1,072</td>
<td>7</td>
<td>100</td>
<td>0.0000</td>
</tr>
<tr>
<td>Total/average</td>
<td>129,671</td>
<td>118.8</td>
<td>15,419</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td>–0.1694</td>
</tr>
</tbody>
</table>

Source: Authors.
**Case in which within-group variances are unknown**  In many applications, the within-group variances will be unknown. For example, the data might have been obtained from published tabulations by income quintile. In such cases, it must be assumed that there is no within-group variance and the second term in equation 8.6 is set to zero. However, in addition, \( n \) needs to be replaced by \( T \) in the denominator of the first term because there are in effect only \( T \) observations, not \( n \).

Table 8.2 gives an example using data on under-five mortality (rates per birth, not rates per 1,000 births) from the 1998 Vietnam Living Standards Survey (VLSS). The data were computed directly from the survey, with children being grouped into household per capita consumption quintiles. The assumption made in table 8.2 is that the within-group variances in mortality are not known and are set to zero. Below, we relax this assumption. The table, which is extracted from an Excel file, shows the values for each quintile of \( R \), \( q \), \( a \), and \( f a^2 \) computed by substituting estimates for the parameters in the formula above. Also shown is the sum of \( fa^2 \) across the five quintiles. Substituting \( \Sigma f a^2 = 0.680 \), \( C = -0.184 \), and \( T = 5 \) into equation 8.6 gives 0.0029 for the variance of the estimate of \( C \) and hence a standard error equal to 0.0537. The \( t \)-statistic for \( C \) is therefore –3.43.

**Case in which within-group variances are known**  In some cases, the within-group variances will be known, and this provides us with more information. In effect, we move from having information only on the \( T \) group means to having information on the full sample—albeit with the variation within the groups being picked up only by the group standard deviations. One such scenario is the case in which we are working with mortality data—the rates are defined at the group level only, but the within-group standard deviations are reported.\(^3\)

In such cases, \( n \) is used (rather than \( T \)) in the denominator of the first term in equation 8.6, and the second term needs to be computed as well. Table 8.3 shows the standard errors for each quintile’s under-five mortality rate from the Vietnam data. The final column shows the value for each quintile of the term in the summation operator in the second term of equation 8.6, as well as the sum of these across the five quintiles. Dividing this sum through by \( n \) gives 1.511e-6, which is the second term of equation 8.6. Dividing \( \Sigma fa^2 \) through by \( n (=5,315) \) gives 2.717e-6, which is

**Table 8.2** Under-Five Deaths in Vietnam, 1989–98 (within-group variance unknown)

<table>
<thead>
<tr>
<th>Consumption group</th>
<th>No. of births</th>
<th>Cumul % births</th>
<th>( R )</th>
<th>U5MR</th>
<th>Cumul % deaths</th>
<th>CI</th>
<th>( q )</th>
<th>( a )</th>
<th>( f a^2 )</th>
</tr>
</thead>
<tbody>
<tr>
<td>Poorest</td>
<td>1,002</td>
<td>19</td>
<td>0.094</td>
<td>0.060</td>
<td>31</td>
<td>–0.024</td>
<td>0.312</td>
<td>0.648</td>
<td>0.079</td>
</tr>
<tr>
<td>2nd</td>
<td>949</td>
<td>37</td>
<td>0.278</td>
<td>0.034</td>
<td>48</td>
<td>–0.013</td>
<td>0.482</td>
<td>0.959</td>
<td>0.164</td>
</tr>
<tr>
<td>Middle</td>
<td>1,002</td>
<td>56</td>
<td>0.461</td>
<td>0.041</td>
<td>69</td>
<td>–0.053</td>
<td>0.695</td>
<td>0.944</td>
<td>0.168</td>
</tr>
<tr>
<td>4th</td>
<td>1,082</td>
<td>76</td>
<td>0.657</td>
<td>0.028</td>
<td>85</td>
<td>–0.095</td>
<td>0.854</td>
<td>0.842</td>
<td>0.144</td>
</tr>
<tr>
<td>Richest</td>
<td>1,280</td>
<td>100</td>
<td>0.880</td>
<td>0.022</td>
<td>100</td>
<td>0.000</td>
<td>1.000</td>
<td>0.719</td>
<td>0.124</td>
</tr>
<tr>
<td>Total/average</td>
<td>5,315</td>
<td></td>
<td>0.036</td>
<td></td>
<td>–0.184</td>
<td></td>
<td>0.680</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

*Source: Authors.*

\(^3\)Or the standard errors of the group means are reported, from which estimates of the variances can be recovered provided group sizes are known.
the first term in equation 8.6. The sum of the two terms is the variance, equal in this case to 4.228e-6, giving a standard error of the estimate of $C$ equal to 0.0021. This, unsurprisingly, is substantially smaller than the standard error obtained assuming no within-group variance.

### Estimation and inference for microdata

#### Point estimate of the concentration index

The concentration index ($C$) can be computed very easily from microdata by using the “convenient covariance” formula (equation 8.3). If the sample is not self-weighted, weights should be applied in computation of the covariance, the mean of the health variable, and the fractional rank. Given the relationship between covariance and ordinary least squares (OLS) regression, an equivalent estimate of the concentration index can be obtained from a “convenient regression” of a transformation of the health variable of interest on the fractional rank in the living standards distribution (Kakwani, Wagstaff, and van Doorslaer 1997). Specifically,

\[
2\sigma_r^2 \left( \frac{h_i}{\mu} \right) \approx \alpha + \beta h_i + \epsilon_{ir}
\]

where $\sigma_r^2$ is the variance of the fractional rank. The OLS estimate of $\beta$ is an estimate of the concentration index equivalent to that obtained from equation 8.3. This method gives rise to an alternative interpretation of the concentration index as the slope of a line passing through the heads of a parade of people, ranked by their living standards, with each individual’s height proportional to the value of his or her health variable, expressed as a fraction of the mean.

### Computation of the concentration index

To illustrate computation, we estimate the concentration index for the public subsidy to hospital outpatient care (subsidy) in Vietnam using data from the 1998 Vietnam Living Standards Survey (see chapter 14 and O’Donnell et al. [2007]). The living standards variable is household consumption per equivalent adult ($eqcons$), and the sample must be weighted (by variable weight).

By using equation 8.3 or 8.7, an estimate of a concentration index can be computed easily with any statistical package. The only slight complication involves com-
putation of the fractional rank variable in the case that the data must be weighted. The weighted fractional rank is defined as follows:

\[ r_i = \sum_{j=0}^{i-1} w_j + \frac{w_i}{2}, \]

where \( w_i \) is the sample weight scaled to sum to 1, observations are sorting in ascending order of living standards, and \( w_0 = 0 \). In Stata, this can be computed as follows:

```stata
gen raw_rank=rank(eqcons), unique
sort raw_rank
quietly sum weight
gen wi=weight/r(sum)
gen cusum=sum(wi)
gen wj=cusum[_n-1]
replace wj=0 if wj==.
gen rank=wj+0.5*wi
```

where `income` is the measure of living standards, `weight` is the original sample weight, `wi` is a scaled version of this that sums to 1, `wj` is the first term in equation 8.8, and `rank` is \( r_i \) in equation 8.8. Alternatively, the weighted fractional rank can be generated using `glcurve`,

```stata
glcurve eqcons [aw=weight], pvar(rank) nograph
```

Because weights are applied, the generated variable `rank` is the weighted fractional rank. The rank variables produced by the two procedures will be perfectly correlated. The (weighted) mean of the rank produced from the first procedure will always be exactly 0.5, and the mean of the rank produced by `glcurve` will differ from 0.5 only at the 4th–5th decimal place.

By using equation 8.3, the concentration index can then be computed using

```stata
qui sum subsidy [fw=weight]
scalar mean=r(mean)
cor subsidy rank [fw=weight], c
sca c=(2/mean)*r(cov_12)
sca list c
```

By using equation 8.7, the index is computed as follows:

```stata
qui sum rank [fw=weight]
sca var_rank=r(Var)
gen lhs=2*var_rank*(subsidy/mean)
regr lhs rank [pw=weight]
sca c=_b[rank]
sca list c
```

4See chapter 7 for an explanation of `glcurve`.
5For the `corr` and `sum` commands, frequency weights (`fw`) should be used to get the correct variance of the weighted rank and its covariance with the health variable of interest. The weight variable must be an integer for the frequency weight to be accepted by Stata. If the weight variable is a noninteger, analysts will first need to multiply the weight by \( 10^k \), where \( k \) is the largest number of decimal places in any value of the weight variable. A new integer weight variable will then have to be created using the `gen new_weight = int (weight)`. The alternative is to use analytic weights and accept some imprecision.
Both procedures give an estimate of the concentration index of 0.16700, indicating that the better-off receive more of the public subsidy to hospital outpatient care in Vietnam.

**Standard error**

Kakwani, Wagstaff, and van Doorslaer (1997) derived the standard error of a concentration estimated from microdata. They did this by noting that the concentration index can be written as a nonlinear function of totals, and so the delta method (Rao 1965) can be applied to obtain the standard error. The resulting formula is essentially a simplified version of equation 8.6 without the second term because at the individual level there is no within-group variation. Specifically,

\[
\text{var} \left( \hat{C} \right) = \frac{1}{n} \left[ \frac{1}{n} \sum_{i=1}^{n} \left( \frac{a_i^2}{\mu} \right) \right] - \left( 1 + C \right)^2,
\]

where

\[
a_i = \frac{h_i (2r_i - 1 - C) + 2 - q_{i-1} - q_i}{\mu},
\]

and

\[
q_i = \frac{1}{\mu n} \sum_{j=1}^{i} h_j,
\]

is the ordinate of the concentration curve \( L_h(p) \), and \( q_0 = 0 \).

Unfortunately, equation 8.9 does not take into account sample weights and other sample design features, such as cluster sampling (see chapters 2 and 9), although in principle it could be adapted to do so. The formula can be computed easily in Stata. We demonstrate this for the Vietnam subsidy example. First, we must recompute the concentration index without the application of weights, as follows:

```
glcurve subsidy, sortvar(eqcons) pvar(ranku) glvar(ccurve) lorenz nograph;
qui sum ranku
sca var_ranku=r(Var)
qui sum subsidy
sca meanu=r(mean)
gen lhsu=2*var_ranku*(subsidy/meanu)
regr lhsu ranku
sca conindu=_b[rank]
```

The estimate of the concentration index from the unweighted data is 0.16606. Standard errors are then computed using equation 8.9 as follows:

```
sort ranku
gen cclag = ccurve[_n-1]
replace cclag=0 if cclag==.
gen asqr=((subsidy/meanu)*(2*ranku-1-conindu)+2-cclag-ccurve)^2
qui sum asqr
sca var=(r(mean)-(1+conindu)^2)/r(N)
sca se=sqrt(var)
sca list conindu se
```

That gives a standard error of 0.033976, and so a \( t \)-ratio of 4.89.
The Concentration Index 103

The limitation of equation 8.9 is that it cannot be applied directly to data that are weighted and/or do not have a simple random sample design. To take such sample features into account, one option is simply to use the standard error of the coefficient on the rank variable in the convenient regression. Because this coefficient is an estimate of the concentration index, one might expect its standard error to be that of the concentration index. This is not quite correct because it takes no account of the sampling variability of the estimate of the mean of the health variable that enters the transformation giving the left-hand side of the convenient regression. Note that the variance of the fractional rank, which is also used in the transformation, depends only on the sample size and so has no sampling variability. It can be treated as a constant. One computationally simple way of taking account of the sampling variability of the mean is to run the convenient regression without transforming the left-hand-side variable but (equivalently) transforming the rank coefficient instead. A delta method standard error can then be computed for the transformed coefficient that takes account of the sampling variability of all terms used in the transformation. From the regression

\[ h_t = \alpha_1 + \beta_1 r_t + u_t \]  

the estimate of the concentration index is given by

\[ \hat{\beta} = \left( \frac{2\sigma^2}{\bar{r}} \right) \hat{\beta}_t. \]  

By using the facts that the least squares predicted value has the same mean as the dependent variable and that the mean of the fractional rank variable is 0.5, equation 8.11 can be written as

\[ \hat{\beta} = \left( \frac{2\sigma^2}{\bar{r} + \frac{\hat{\beta}_t}{2}} \right) \hat{\beta}_t. \]

Because the estimate is now written as a function of the regression coefficients, a standard error can be obtained by applying the delta method. In Stata, this procedure can be implemented very easily using `nlcom`.

\[ \text{regr subsidy rank [pw=weight]} \]  
\[ \text{nlcom ((2*var_rank)/(_b[_cons]+0.5*_b[rank]))*_b[rank]} \]

For the Vietnam outpatient subsidy example, that gives a standard error of 0.034016 for the estimate of the (weighted) concentration index of 0.16700 reported above. The standard error of the rank coefficient from equation 8.7 is 0.034945, and so it appears that taking account of the sampling variability of the mean makes very little difference. Experimentation suggests that this is generally the case, and so standard errors from the convenient regression equation 8.7 can be used without too much concern for inaccuracy.

In Stata, if weights are applied in the regression, then the standard error returned will be robust to heteroskedasticity. If there are no weights, heteroskedasticity robust

---

6This is due to the nature of the fractional rank variable. Its weighted mean is always 0.5, and its variance approaches 1/12 as n goes to infinity. For given n, the variance of the fractional rank is always the same.
standard errors can be obtained by adding the option robust to the regression. If this is done, then the delta method standard errors computed by a `nlcom` command following the regression will also be robust. If the survey has a cluster sampling design, then the standard errors should be corrected for within-cluster correlation. This is achieved by adding the option `cluster`,

```
regr subsidy rank [pw=wt], cluster(commune)
nlcom ((2*var_rank)/(_b[_cons]+0.5*_b[rank]))*_b[rank]
```

where commune is the variable denoting the primary sampling unit—communes in the VLSS. Allowing for within-cluster correlation raises the standard error in the Vietnam subsidy example from 0.034016 to 0.041988.

Correcting for across-cluster correlation may or may not be necessary, depending on the sample design, but a form of serial correlation is always likely to be present owing to the rank nature of the regressor (Kakwani, Wagstaff, Van Doorslaer 1997). To correct the standard errors for this, one can use the Newey-West (Newey and West 1994) variance-covariance matrix, which corrects for autocorrelation, as well as heteroscedasticity. In Stata, the command `newey` produces OLS regression coefficients with Newey-West standard errors. To use this, the data must be set to a time series format with the time variable being, in this case, the living standards rank. This must be an integer valued variable, and so the fractional rank created above cannot be used. Below, we create the appropriate rank variable (`ranki`) before running the `newey` command:

```
egen ranki=rank(eqcons), unique
tset ranki
newey subsidy rank [aw=weight], lag(1)
nlcom ((2*var_rank)/(_b[_cons]+0.5*_b[rank]))*_b[rank]
```

Note that the (weighted) fractional rank and not the integer valued rank is still used in the regression. Weights (analytical) are allowed, and the `lag(#)` option must be included to specify the maximum number of lags to be considered in the autocorrelation structure. For our example, this estimator gives a standard error of 0.034568, slightly larger than if we allow for heteroskedasticity only (0.034016), but smaller than if we allow for within-cluster correlation (0.041988).

### Demographic standardization of the concentration index

As discussed in chapter 5, we are often interested in measuring socioeconomic-related inequality in a health variable after controlling for the confounding effect of demographics. In chapter 5 we explained how this can be done using both direct and indirect methods of standardization. To estimate a standardized concentration index, one could use either method of standardization to generate a predicted health variable purged of the influence of demographics across socioeconomic groups, as explained in chapter 5, then compute the concentration index for this standardized variable.

In the case that one wishes to standardize for the full correlation with confounders, and so there are no control (z) variables (see chapter 5), a shortcut method of obtaining an indirectly standardized concentration index is simply to include the standardizing variables directly in the convenient regression. This is precisely
what is being done in the literature that makes use of the relative index of inequality (e.g., Mackenbach et al. [1997]). From the regression

\[
2 \sigma^2 \left( \frac{h_i}{\mu} \right) = \alpha + \beta_2 r_i + \sum_j \delta_j x_{ij} + u_i,
\]

where \( x_i \) are the confounding variables, for example, age, sex, and so on, the OLS estimate \( \hat{\beta}_2 \) is an estimate of the indirectly standardized concentration index. Computation requires simply adding the confounding variables to the regression commands discussed above.

**Sensitivity of the concentration index to the living standards measure**

In chapter 6 we described alternative measures of living standards—consumption, expenditure, wealth index—and noted that it is not always possible to establish a clear advantage of one measure over others. It is therefore important to consider whether the chosen measure of living standards influences the measured degree of socioeconomic-related inequality in the health variable of interest. When the concentration index is used as a summary measure of inequality, the question is whether it is sensitive to the living standards measure.

As noted above, the concentration index reflects the relationship between the health variable and living standards rank. It is not influenced by the variance of the living standards measure. In some circumstances, this may be considered a disadvantage. For example, it means that, for a given relationship between income and health, the concentration index cannot discriminate the degree of income-related health inequality in one country in which income is distributed very unevenly from that in another country in which the income distribution is very equal. On the other hand, when one is interested in inequality at a certain place and time, it is reassuring that the differing variances of alternative measures of living standards will not influence the concentration index. However, the concentration index may differ if the ranking of individuals is inconsistent across alternative measures.

Wagstaff and Watanabe (2003) demonstrate that the concentration index will differ across alternative living standards measures if the health variable is correlated with changes in an individual’s rank on moving from one measure to another. The difference between two concentration indices \( C_1 \) and \( C_2 \), where the respective concentration index is calculated on the basis of a given ranking \( (r_1, r_2) \)—for example, consumption and a wealth index—can be computed by means of the regression

\[
2 \sigma^2 \left( \frac{h_i}{\mu} \right) = \alpha + \gamma \Delta r_i + \epsilon_i,
\]

where \( \Delta r_i = r_{1i} - r_{2i} \) is the reranking that results from changing the measure of socioeconomic status, and \( \sigma^2 \) is its variance. The OLS estimate of \( \gamma \) provides an estimate of the difference \( (C_1 - C_2) \). Significance of the difference between indices can be tested by using the standard error of \( \gamma \).

For 19 countries, Wagstaff and Watanabe (2003) test the sensitivity of the concentration index for child malnutrition to the use of household consumption and a

---

7This ignores the sampling variability of the left-hand-side estimates.
wealth index as the living standards ranking variable. Malnutrition is measured by a binary indicator of underweight and another for stunting (see chapter 4). For each of underweight and stunting, the difference between the concentration indices is significant (10%) for 6 of 19 comparisons. This suggests that in the majority of countries, child nutritional status is not strongly correlated with inconsistencies in the ranking of households by consumption and wealth.

But there is some evidence that concentration indices for health service utilization are more sensitive to the living standards measure. Table 8.4, reproduced from Lindelow (2006), shows substantial and significant differences between the concentration indices (CI) for a variety of health services in Mozambique using consumption and an asset index as the living standards measure. In the case of consumption, the concentration index indicates statistically significant inequality in favor of richer households for all services. With households ranked by the asset index rather than consumption, the inequality is greater for all services except health center visits, for which the concentration index indicates inequality in utilization in favor of poorer households.

It appears that the choice of welfare indicator can have a large and significant impact on measured socioeconomic inequalities in a health variable, but it depends on the variable examined. Differences in measured inequality reflect the fact that consumption and the asset index measure different things, or at least are different proxies for the same underlying variable of interest. But only in cases in which the difference in rankings between the measures is also correlated with the health variable of interest will the choice of indicator have an important impact on the findings. In cases in which both asset and consumption data are available, analysts are in a position to qualify any analysis of these issues by reference to parallel analysis based on alternative measures. However, data on both consumption and assets are often not available. In these cases, the potential sensitivity of the findings should be explicitly recognized.

References


